

Dallamcsengő mikroprocesszorral

Kinczel András

Bár már többször jelent meg dallamcsengő leírása a Rádiótechnika hasábjain, most mégis egy újabb ilyen áramkört ismertetünk. Az eddig megjelent kapcsolások általában csak egyetlen és viszonylag rövid dallam lejátszására voltak képesek, több dallamból való választási lehetőség pedig csak sok IC és még több diszkrét elem (dióda, potenciométer stb.) felhasználásával volt elérhető.


Áramkörünk elkészítéséhez a személyi számítógépek zenélési képessége adta az ötletet. Mikroprocesszor felhasználásával rendkívül egyszerű áramkör készíthető (1. ábra), mely képes különböző hangok előállítására. A megfelelő dallamok megszóltatása már csak az alkalmazott programtól függ.

A dallamcsengő elvi kapcsolási rajza 2. ábrán látható. Az áramkör öt IC-ből felépített mikroprocesszoros kapcsolás, amelyben processzorként Z80-at alkalmaztunk. A processzor számára az órajelket egy 74LS132 típusú 4×2 -bemenetű NAND Smitt-trigger két kapuja állítja elő, amely RC oszcillátorként működik. Az órajel frekvenciája az 1,2 nF-os kondenzátorral, vagy a 330 Ω -os ellenállással befolyásolható. A tapasztalatok azt mutatják, hogy az RC elemek pontos értéke a hangképzés szempontjából nem túl kritikus, de szükség esetén a 330 Ω -os ellenállás potenciométerrel való helyettesítésével a megfelelő óra-frekvencia „fűlre” beállítható. Ilyenkor a hangskála a megfelelő helyre


kerül, mint pl. a normál „á” 440 Hz értéket vesz fel.

A 2716 típusjelű EPROM tárolja a vezérlő program utasításkódjait és a dallamoknak megfelelő programtáblázatot. Ebben minden hangnak egy-egy bájt felel meg. A processzor negált kimenetű IORQ pontján megjelenő és a dallamnak megfelelő ütemben érkező impulzusokat egy D-flip-flop alakítja át négyszögjelekké (74LS74). Ezek a négyszögjelek egy TBA810 típusú IC-vel felépített erősítőre kerülnek. Ennek teljesítménye bőségesen elegendő akár egy nagyméretű hangszóró meghajtásához is. A szükséges hangerő a 10 k Ω -os potenciométerrel állítható be.

A tápláláshoz szükséges tápegység a 3. ábrán látható. A relatív kicsiny áramfelvé-



1. ábra. A dobozából kiemelt készülék képe. Az előtérben UV-lámpával törölhető EPROM-ok láthatók



4 1/2 digit panelmérő OE-135

Műszaki adatok:

Táp feszültség:	+5 V \pm 0,25 V DC
Táparám:	\leq 100 mA
Mérési határ:	\pm 2.0000 V
Túlcsordulás:	villogással jelzi
Bemenő ellenállás:	\geq 10 ³ M Ω
Differenciál linearitás:	0,01 LSB
Linearitás \pm teljes skála:	\pm 1 digit
Minden kimenete:	TTL kompatibilis
Igazi differenciál bemenet	
Működési hőmérséklet:	0 – +60 C°
Méretei:	100 x 42 x 85 mm


Alkalmazási terület:

Analóg feszültség digitális kijelzésére alkalmazható.
Hat kiegészítő be- és kimenete lehetővé teszi UART-hoz, mikroprocesszorhoz, vagy más komplex áramkörhöz való csatlakoztatását.
Multiplexelt BCD kimenettel rendelkezik.


Gyártja: Orvosi Műszer Szövetkezet
1081 Budapest, Rákóczi út 71.
Tel.: 142-642, 339-757

Forgalmazza: MIGÉRT
Exportálja: METRIMPEX
A változtatás joga fenntartva!







2. ábra. A dallamszengő kapcsolási rajza




5. ábra. A nyák-lemez alkatrész-oldali fólia-rajzolata



3. ábra. A tápegység kapcsolási rajza



4. ábra. A nyák-lemez fólia-oldali rajzolata



6. ábra. A nyák-lemez beültetési rajza

tel következtében hálózati transzformátor-ként csengőreduktor is szóba jöhet. Így az esetleges régi csengő transzformátora ezen új megoldásnál is felhasználható.

Az áramkör nyomtatási terve nagy mértékben függ a felhasznált alkatrészek méretétől, illetve a NYÁK-készítés technológiájától (egy-, vagy kétoldalas kivétel), így azt az adott lehetőségek figyelembevételével célszerű kialakítani. A kis számú alkatrész miatt sikerrel alkalmazható a huzalozásos technika is. A 4. és 5. ábrán egy kétoldalt foliozott nyák-tervet mutatunk be. Ennek alkatrész-beültetése a 6. ábrán látható.

A vezérlő programot az 1. Táblázat szemlélteti. A processzor kétféle megszakítási lehetőségét kihasználva, a program (a kapcsoló állásától függően) kétféle üzemmódra ad lehetőséget.

1. Táblázat A VEZÉRLŐ PROGRAM

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
0000	ED	56	21	9F	00	F8	AF	76	23	47	57	7E	5F	E6	1F	F6
0010	80	4F	7B	07	07	07	E6	07	F6	30	5F	1A	57	0A	FE	00
0020	28	19	FE	FE	28	DF	FE	FF	28	D8	4F	AF	5F	C3	46	00
0030	FF	C1	81	61	41	31	21	11	C3	08	00	06	01	10	FE	18
0040	8A	20	F8	C3	08	00	41	18	8A	28	8D	10	FA	D3	01	C3
0050	46	00	16	80	06	24	18	8A	28	06	10	FA	D3	01	18	F4
0060	06	FF	10	FE	18	9F	16	20	06	20	18	8A	28	06	10	FA
0070	D3	01	18	F4	06	FF	10	FE	C3	52	00	FF	FF	FF	FF	FF
0080	00	11	12	13	14	15	16	18	19	1B	1D	1E	20	22	24	26
0090	29	28	2E	31	34	36	3A	3D	41	45	4A	4D	51	56	FE	FF

2. táblázat A program hangválasztéka

magasság	hosszúság							
	E	3/4	1/2	3/8	1/4	3/16	1/8	1/16
SZUNET	00	20	40	60	80	A0	C0	E0
H	01	21	41	61	81	A1	C1	E1
AISZ	02	22	42	62	82	A2	C2	E2
A	03	23	43	63	83	A3	C3	E3
GISZ	04	24	44	64	84	A4	C4	E4
G	05	25	45	65	85	A5	C5	E5
FISZ	06	26	46	66	86	A6	C6	E6
F	07	27	47	67	87	A7	C7	E7
E	08	28	48	68	88	A8	C8	E8
DISZ	09	29	49	69	89	A9	C9	E9
D	0A	2A	4A	6A	8A	AA	CA	EA
CISZ	0B	2B	4B	6B	8B	AB	CB	EB
C	0C	2C	4C	6C	8C	AC	CC	EC

H	0D	2D	4D	6D	8D	AD	CD	ED
AISZ	0E	2E	4E	6E	8E	AE	CE	EE
A (normál A)	0F	2F	4F	6F	8F	AF	CF	EF
GISZ	10	30	50	70	90	B0	D0	F0
G	11	31	51	71	91	B1	D1	F1
FISZ	12	32	52	72	92	B2	D2	F2
F	13	33	53	73	93	B3	D3	F3
E	14	34	54	74	94	B4	D4	F4
DISZ	15	35	55	75	95	B5	D5	F5
D	16	36	56	76	96	B6	D6	F6
CISZ	17	37	57	77	97	B7	D7	F7
C	18	38	58	78	98	B8	D8	F8

H	19	39	59	79	99	B9	D9	F9
AISZ	1A	3A	5A	7A	9A	BA	DA	FA
A	1B	3B	5B	7B	9B	BB	DB	FB
GISZ	1C	3C	5C	7C	9C	BC	DC	FC
G	1D	3D	5D	7D	9D	BD	DD	FD

DAL VEG								FE
DALSOR VEG								FF

A kapcsoló 1. állásában a nyomógomb megnyomására a dallamsorból mindig a soronkövetkező kerül lejátszásra. Az egyes hangoknak megfelelő bájtokat a vezérlő-program végén az A0 címtől kezdődően kell elhelyezni. Egy bájtkban 3 bit hordozza a hang hosszúságát, öt biten pedig a hang magasságát adjuk meg. Így 8-féle hosszúságban 30 féle hangmagasság állítható elő, beleértve a szünetet is. A hangok megadásának módját a 2. Táblázat mutatja.

A fennmaradó bájtkok közül kettő a dallam, illetve a dallamsor végét jelzi. Egy-egy dallam utolsó hangját FE-nek kell követnie; a program innen szerez tudomást ar-

ról, hogy a dallam véget ért. Az utolsó dallam után nem kell FE-t írunk, az üres EPROM rekeszt észlelve (FF) a program az első dallam kezdő címére. A0-ra ugrik vissza. Az EPROM memória EPROM-égetővel írható be.

Egy-egy dallam, hosszúságától függően, kb. 20-30 bájtkot foglal el. Így a 2716 kapacitását figyelembe véve megközelítőleg 150 dallamból álló sort tudunk előállítani. Példaképpen a 3. Táblázatban megadjuk néhány ismert dallam adatait.

A kapcsoló 2. állásában a nyomógomb a processzor nem-maszkolható interrupt (NMI negált) lábára kerül. Ekkor a készü-

lék a bim-bam csengőhöz hasonló módon két hangból álló jelzést ad. Két egymástól független nyomógomb csatlakoztatása esetén lehetőség nyílik a készülék több helyről történő működtetésére is. Ilyenkor az egyik nyomógomb pl. bim-bam üzemen működteti a készüléket, míg a másik nyomógombról a készülék dallamokat játszik.

3. Táblázat A DALLAMOK HANGKÉSZLETE

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
00A0	34	72	71	52	84	34	34	76	89	B6	14	FE	0C	AD	AC	AA
00B0	AC	AD	AF	6C	AC	AF	0C	AD	AC	AF	B1	84	B3	31	31	B1
00C0	83	84	86	84	83	81	AF	31	31	B1	AF	AD	AF	B1	83	B4
00D0	86	84	86	38	88	86	74	33	36	11	FE	98	53	93	53	8F
00E0	51	93	51	8F	53	93	4F	8C	0A	C0	8A	4C	3F	4F	93	51
00F0	93	51	CF	D1	53	96	56	98	13	FE	56	92	91	2F	8F	8D

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
0100	8A	80	8D	2F	88	56	92	91	2F	8F	8D	8F	91	92	34	80
0110	FE	8C	ED	6C	6F	68	8C	ED	6C	6F	68	8C	ED	6C	6F	68
0120	6C	AD	8E	FE	8D	8C	4A	0E	4F	8D	36	8D	8C	4A	4D	4D
0130	0E	4F	8D	36	92	94	16	FE	51	8F	8D	4C	8D	8F	51	8F
0140	8D	4C	8D	8F	51	8F	8D	4C	8D	8F	51	4F	4F	FE	4C	6A
0150	CC	4F	4F	6F	D1	6F	CE	2F	88	4E	71	CC	2F	88	53	76
0160	D1	38	FE	91	91	8F	6D	CC	8A	8A	8A	88	4A	8D	4C	8D
0170	2F	91	52	31	FE	D1	88	88	8A	8A	6C	CC	8D	CC	CC	8C
0180	8C	8D	8D	6F	CF	91	CC	D1	94	8C	8D	8F	4A	52	91	FE
0190	51	91	8F	91	93	94	98	56	96	93	94	96	58	FE	56	96
01A0	93	94	96	91	91	56	96	93	94	96	58	FE	65	CC	CC	CC
01B0	87	87	CC	CA	68	CC	CC	8A	8A	CC	CA	6F	F3	F1	CF	CF
01C0	D3	8D	8D	CC	D1	76	F6	F4	D3	D6	91	71	FE	D1	CF	CC
01D0	CC	D1	CF	CC	CC	CA	CC	C7	CC	8A	85	4A	8A	FE	84	73
01E0	84	78	84	B3	B3	84	76	74	68	84	B3	74	78	84	86	86
01F0	88	79	78	FE	51	53	4D	8C	87	4A	8C	87	4A	4D	4C	51

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
0200	53	FE	91	93	8F	56	56	91	8F	8D	8F	8C	FE	99	99	98
0210	96	96	98	99	98	9D	9D	98	99	79	D8	58	FE	78	D6	D4
0220	D6	D4	D6	91	91	56	79	D8	D6	D8	D6	D9	98	96	D8	D9
0230	D8	D4	FE	8F	8D	8C	8A	CC	CC	CF	D1	8F	2F	91	8F	8F
0240	8D	4D	8D	8F	8F	91	8F	FE	8C	8C	4A	8C	8F	4D	8F	93
0250	91	94	53	53	FE	8C	91	D4	F4	F3	91	8C	91	D4	F6	F4
0260	98	FE	D8	D4	D4	D1	EF	F1	F3	F4	96	D8	D4	D4	D1	EF
0270	F1	F3	F4	96	F8	F6	D4	B6	F8	98	98	FF	FF	FF	FF	FF


```

1 ;
2 ; Z80 dallamcsengő eredeti vezérlőprogramjának visszafejtése.
3 ; (Ez nincs kifejtve a cikkben.)
4 ;
5 ; Publikálta: Kinczel András, a Rádiótechnika 1987/2 számában.
6 ; Dallamcsengő mikroprocesszorral (99.-101. oldalakon)
7 ;
8 ; Szerző: TCSabI©
9 ; Dátum: 2017. július 27.
10 ; E-mail: toth.csaba.hip@gmail.com
11 ;
12 0000 ED56 IM 1 ; A CPU Interrupt Mode 1-be állítása.
13 ; Az INT bemenétre egy RST 38H
14 ; (azaz CALL 0038H) utasítást fog végrehajtani.
15 ; Az NMI bemenétre egy RST 66H
16 ; (azaz CALL 0066H) utasítást fog végrehajtani.
17 ; Az INT és NMI bemeneteken vannak,
18 ; a dallamindító gombok.
19 ; Visszatérési pont, a dallamok ismételt lejátszásához.
20 0002 219F00 LD HL,009FH ; HL = 009FH
21 ; A dallamok kezdőcímeinek inicializálása.
22 ; (Azért nem 00A0h, mert a HALT után van egy
23 ; INC HL utasítás.)
24 ; Visszatérési pont, a következő dallam lejátszásához.
25 0005 FB EI ; Megszakítások engedélyezése.
26 0006 AF XOR A ; A = A xor A, azaz Akkumulátor nullázása 00h.
27 0007 76 HALT ; CPU működésének felfüggesztése,
28 ; a következő megszakítás kérésig.
29 ; Azaz várakozás, valamelyik gomb megnyomására.
30 ; Visszatérési pont, a következő hang lejátszásához.
31 0008 23 INC HL ; HL = HL + 1
32 0009 47 LD B,A ; B = A
33 000a 57 LD D,A ; D = A
34 000b 7E LD A,(HL) ; A = (HL), A dallam hangjának kiolvasása
35 ; a memóriából.
36 000c 5F LD E,A ; E = A, Dallam hangjának mentése E-be.
37 000d E61F AND 1FH ; A = A and 1Fh 00011111 Frekvencia kimaszkolása.
38 000f F680 OR 80H ; A = A or 80h 10000000 Így lesz frekvencia cím.
39 0011 4F LD C,A ; C = A, Így BC regiszterben van a frekvencia.
40 0012 7B LD A,E ; A = E, A dallam hangjának kiolvasása E-ből.
41 0013 07 RLCA ; A eltolása balra 1 bittel.
42 0014 07 RLCA ; A 7.bit -> 0.bitbe töltődik.
43 0015 07 RLCA ;
44 0016 E607 AND 07H ; A = A and 07h 00000111 Hanghossz kimaszkolása.
45 0018 F630 OR 30H ; A = A or 30h 00110000 Így lesz hanghossz cím.
46 001a 5F LD E,A ; E = A, Így a DE regiszterben van a hang hossza.
47 001b 1A LD A,(DE) ; A = (DE), A hang hosszának kiolvasása,
48 ; a táblázatból.
49 001c 57 LD D,A ; D = A
50 001d 0A LD A,(BC) ; A = (BC), A frekvencia kiolvasása,
51 ; a táblázatból.
52 001e FE00 CP 00H ; A == 00h?, azaz A-00h --> Z bit beállítása.
53 0020 2819 JR Z,003BH ; Relatív ugrás a 003Bh címre, ha Z=1. Szünet...
54 0022 FEFE CP 0FEH ; A == FEh?, azaz A-FEh --> Z bit beállítása.
55 0024 28DF JR Z,0005H ; Relatív ugrás a 0005h címre, ha Z=1.
56 ; Következő dallam lejátszása.
57 0026 FEFF CP 0FFH ; A == FFh?, azaz A-FFh --> Z bit beállítása.
58 0028 28D8 JR Z,0002H ; Relatív ugrás a 0002h címre, ha Z=1.
59 ; Dallamkészlet ismételt lejátszása.
60 002a 4F LD C,A ; C = A
61 002b AF XOR A ; A = A xor A, azaz Akkumulátor nullázása 00h.
62 002c 5F LD E,A ; E = A
63 002d C34600 JP 0046H ; Ugrás a 0046h címre. Hang generálása...
64 ;-----
65 ; A hangok hosszúságának táblázata.
66 ;-----
67 0030 FF ; Egész
68 0031 C1 ; 3/4
69 0032 81 ; 1/2
70 0033 61 ; 3/8
71 0034 41 ; 1/4
72 0035 31 ; 3/16
73 0036 21 ; 1/8

```

```

74 0037 11 ; 1/16
75 ;-----
76 ; A CPU INT bemenetére megszakításkérés érkezett.
77 ; A dallamcsengő INT bementén lévő gombját megnyomták. Dallam lejátszása.
78 0038 C30800 JP 0008H ; Ugrás a 0008h címre. Következő hang lejátszása.
79 ; Hang nélküli várakozás. Azaz szünet lejátszása... DE időtartammal.
80 003b 0601 LD B,01H ; B = 01h
81 ;*
82 003d 10FE DJNZ 003DH ; B = B - 1, ha B nem lett 0,
83 ; akkor relatív ugrás a 003Dh címre.
84 003f 1B DEC DE ; DE = DE - 1
85 0040 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
86 0041 20F8 JR NZ,003BH ; Relatív ugrás a 003Bh címre, ha Z=0.
87 0043 C30800 JP 0008H ; Ugrás a 0008h címre. Következő hang lejátszása.
88 ; Hang generálása... BC frekvenciával és DE időtartammal.
89 0046 41 LD B,C ; B = C
90 ;
91 0047 1B DEC DE ; DE = DE - 1
92 0048 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
93 0049 28BD JR Z,0008H ; Relatív ugrás a 0008h címre, ha Z=1.
94 ; Következő hang lejátszása.
95 004b 10FA DJNZ 0047H ; B = B - 1, ha B nem lett 0,
96 ; akkor relatív ugrás a 0047h címre.
97 004d D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
98 ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
99 ; A CPU ekkor, a IORQ kimenetére egy túimpulzust
100 ; küld. Az adatbuszon érvényes cím van.
101 ; A túimpulzus, az SN7474-es D tárolóban szim-
102 ; metrikus négyszögjellé alakul. Ezt halljuk.
103 004f C34600 JP 0046H ; Ugrás a 0046h címre.
104 ; A Bim-Bam második hangjának lejátszása...
105 0052 1680 LD D,80H ; D = 80h
106 ;*
107 0054 0624 LD B,24H ; B = 24h
108 ;*
109 0056 1B DEC DE ; DE = DE - 1
110 0057 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
111 0058 2806 JR Z,0060H ; Relatív ugrás a 0060h címre, ha Z=1.
112 ; Szünet lejátszása.
113 005a 10FA DJNZ 0056H ; B = B - 1, ha B nem lett 0,
114 ; akkor relatív ugrás a 0056h címre.
115 005c D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
116 ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
117 ; A CPU ekkor, a IORQ kimenetére egy túimpulzust
118 ; küld. Az adatbuszon érvényes cím van.
119 ; A túimpulzus, az SN7474-es D tárolóban szim-
120 ; metrikus négyszögjellé alakul. Ezt halljuk.
121 005e 18F4 JR 0054H ; Relatív ugrás a 0054h címre.
122 ; Várakozás...
123 0060 06FF LD B,0FFH ; B = FFh
124 ;*
125 0062 10FE DJNZ 0062H ; B = B - 1, ha B nem lett 0,
126 ; akkor relatív ugrás a 0062h címre.
127 0064 189F JR 0005H ; Relatív ugrás a 0005h címre.
128 ; Visszaugrás a dallam lejátszási üzemmódba.
129 ;-----
130 ; A CPU NMI bemenetére megszakításkérés érkezett.
131 ; A dallamcsengő NMI bementén lévő gombját megnyomták.
132 ; A Bim-Bam első hangjának lejátszása...
133 0066 1620 LD D,20H ; D = 20h
134 ;
135 0068 0620 LD B,20H ; B = 20h
136 ;*
137 006a 1B DEC DE ; DE = DE - 1
138 006b BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
139 006c 2806 JR Z,0074H ; Relatív ugrás a 0074h címre, ha Z=1.
140 ; Szünet lejátszása.
141 006e 10FA DJNZ 006AH ; B = B - 1, ha B nem lett 0,
142 ; akkor relatív ugrás a 0062h címre.
143 0070 D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
144 ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
145 ; A CPU ekkor, a IORQ kimenetére egy túimpulzust
146 ; küld. Az adatbuszon érvényes cím van.

```

```

147                                     ; A túimpulzus, az SN7474-es D tárolóban szim-
148                                     ; metrikus négyszögjellé alakul. Ezt halljuk.
149 0072 18F4      JR      0068H      ; Relatív ugrás a 0068h címre.
150 ; Várakozás...
151 0074 06FF      LD      B,0FFH      ; B = FFh
152 ;*
153 0076 10FE      DJNZ     0076H      ; B = B - 1, ha B nem lett 0,
154                                     ; akkor relatív ugrás a 0076h címre.
155 0078 C35200     JP      0052H      ; Ugrás a 0052h címre.
156 ;-----
157 007b FF        ;
158 007c FF        ;
159 007d FF        ; Nem használt terület.
160 007e FF        ;
161 007f FF        ;
162 ;-----
163 ; A hangok frekvenciájának táblázat...
164 ; A frekvencia és a várakozások időtartama, a CPU órajelétől függ!
165 ;-----
166 0080 00        ; Nincs hang, azaz szünet.
167 ;-----
168 0081 11        ; H'                +
169 0082 12        ; Aisz'              +
170 0083 13        ; A'                +
171 0084 14        ; Gisz'             +
172 0085 15        ; G'                +
173 0086 16        ; Fisz'             +
174 0087 18        ; F'                +
175 0088 19        ; E'                +
176 0089 1B        ; Disz'            +
177 008a 1D        ; D'                +
178 008b 1E        ; Cisz'            +
179 008c 20        ; C'                +
180 ;-----
181 008d 22        ; H                +
182 008e 24        ; Aisz             előző hang + 2^(1/12) (1 oktáv 12 hang)
183 008f 26        ; A (normál A hang) 440Hz
184 0090 29        ; Gisz             előző hang - 2^(1/12) (1 oktáv 12 hang)
185 0091 2B        ; G                -
186 0092 2E        ; Fisz             -
187 0093 31        ; E                -
188 0094 34        ; Disz             -
189 0095 36        ; D                -
190 0095 3A        ; Cisz             -
191 0097 3D        ; C                -
192 ;-----
193 0098 41        ; 'H               -
194 0099 45        ; 'Aisz            -
195 009a 4A        ; 'A               -
196 009b 4D        ; 'Gisz            -
197 009c 51        ; 'G               -
198 009d 56        ; 'Fisz            -
199 ;-----
200 009e FE        ; Dallam vége jelző.
201 009f FF        ; Dallamkészlet vége jelző.
202 ;-----
203 ; Innen kezdődnek a dallamok...
204 00A0 .. .. .   ; A tárolt hang szerkezete...
205 ;
206 ; Egy bájt, 1 hang.
207 ; A hang hossza: 7-5 bitek, azaz 3 biten.
208 ; A hang frekvenciája: 4-0 bitek, azaz 5 biten.

```

**Z80 dallamcsengő hangkészletének kódolása
a mai temperált hangrendszerben
(a hangok távolsága 2^{^(1/12)})**

Hangmagasság	Frekvencia Hz-ben 1,059463094	Hang kód 5 bit	Hang hossza							
			1 Egész	0,75 3/4	0,5 1/2	0,375 3/8	0,25 1/4	0,1875 3/16	0,125 1/8	0,0625 1/16
Szünet	0	0	00	20	40	60	80	A0	C0	E0
H	988	1	01	21	41	61	81	A1	C1	E1
Aisz	932	2	02	22	42	62	82	A2	C2	E2
A	880	3	03	23	43	63	83	A3	C3	E3
Gisz	830	4	04	24	44	64	84	A4	C4	E4
G	784	5	05	25	45	65	85	A5	C5	E5
Fisz	740	6	06	26	46	66	86	A6	C6	E6
F	698	7	07	27	47	67	87	A7	C7	E7
E	658	8	08	28	48	68	88	A8	C8	E8
Disz	622	9	09	29	49	69	89	A9	C9	E9
D	588	10	0A	2A	4A	6A	8A	AA	CA	EA
Cisz	554	11	0B	2B	4B	6B	8B	AB	CB	EB
C	522	12	0C	2C	4C	6C	8C	AC	CC	EC
H	494	13	0D	2D	4D	6D	8D	AD	CD	ED
Aisz	466	14	0E	2E	4E	6E	8E	AE	CE	EE
A (Normal A)	440	15	0F	2F	4F	6F	8F	AF	CF	EF
Gisz	415	16	10	30	50	70	90	B0	D0	F0
G	392	17	11	31	51	71	91	B1	D1	F1
Fisz	370	18	12	32	52	72	92	B2	D2	F2
F	349	19	13	33	53	73	93	B3	D3	F3
E	329	20	14	34	54	74	94	B4	D4	F4
Disz	311	21	15	35	55	75	95	B5	D5	F5
D	294	22	16	36	56	76	96	B6	D6	F6
Cisz	277	23	17	37	57	77	97	B7	D7	F7
C	261	24	18	38	58	78	98	B8	D8	F8
H	247	25	19	39	59	79	99	B9	D9	F9
Aisz	233	26	1A	3A	5A	7A	9A	BA	DA	FA
A	220	27	1B	3B	5B	7B	9B	BB	DB	FB
Gisz	208	28	1C	3C	5C	7C	9C	BC	DC	FC
G	196	29	1D	3D	5D	7D	9D	BD	DD	FD
Dallam vége jelző			FE							
Dallamkészlet vége jelző			FF							

* Egyes EPROM tartalmakban, a dallam vége jelző FFh! A dallamkészlet vége jelző szintén FFh!

* Random dallam lejátszáshoz, egy mátrixa össze kell gyűjteni, az összes dallam kezdőcímét.

**Az EPROM dallamkészletének
dekódolása**

A hang frekvenciája és hossza a bájtban							
2^7	2^6	2^5	2^4	2^3	2^2	2^1	2^0
Hanghossz				Hangfrekvencia			
0-7 dec				0-29 dec			

And E0h >> 5 bit

And 1Fh

Hang hossza			
Hossz	3 bit	szorzó	ms
Egész	0	1	1000
3/4	1	0,75	750
1/2	2	0,5	500
3/8	3	0,375	375
1/4	4	0,25	250
3/16	5	0,1875	188
1/8	6	0,125	125
1/16	7	0,0625	63

A hangok lejátszási sebességének változtatásához, egy analóg bemenet is szükséges.

Hang maszkolása		
Érték		
And	00110100	34h
Eredm.	00011111	1Fh
Eredm.	00010100	14h

Kiolvasott érték.

Logikai művelet bitenként.

Ezt kell kikeresni, a frekvencia táblázatból.

Hanghossz maszkolása		
Érték		
And	00110100	34h
Eredm.	11100000	E0h
Eredm.	00100000	20h
>> 5 bit	00000001	01h

Kiolvasott érték.

Logikai művelet bitenként,

Eltolás 5 bittel jobbra.

Ezt kell kikeresni, a hang hossza táblázatból.

(Így könnyen lejátszható az EPROM-ok tartalma, más típusú mikrovezérlővel is...)